/* Nom du fichier : 2. Traitement sortie rupture


               */
/***************************************************************************/
/* Date de création      : 30/10/2013                                      */
/*Date de modification   : 30/10/2013

/***************************************************************************/
/* objet : Création de la base finale CLEA






       */
/***************************************************************************/
/*libname pano 'O:\UDEV\Réservé DGA UDEV\Partenariat DEV-PREV\Panorama 2013\4.Base SAS';*/
libname pano 'D:\...\Pano';

%let base=pano.clea;

%let base1=pano.panofinal;

%include 'O:\...\0. Format Panorama.sas';


/*************************************************************************************************/ 


/**

PARTIE 1  Préparation de bases et création de variables pour les traitements ultérieurs
**/

/*************************************************************************************************/ 

/* I. Glori@ */
data pacte_gloria; set &base1;

length pacte_sitefor $100. pacte_dom $100. pacte_entp $100. ;

/* Création modalités de réponses pour variables pacte */

/*Site de F° */
if CODEPOSTALSITEF in ('78208' '78423' '78688' '78621' '78356' '78644' '78297') then pacte_sitefor ="SQY";

if CODEPOSTALSITEF in ('94022' '94028' '94044' '94011') then pacte_sitefor ="Plaine_centrale_du_VDM";


/*Entreprise du jeune */
if CODEPOSTALENT in ('78208' '78423' '78688' '78621' '78356' '78644' '78297') then pacte_entp ="SQY";

if CODEPOSTALENT in ('94022' '94028' '94044' '94011') then pacte_entp ="Plaine_centrale_du_VDM";


/*Domicile du jeune */
if CODEPOSTAL in ('78208' '78423' '78688' '78621' '78356' '78644' '78297') then pacte_dom ="SQY";

if CODEPOSTAL  in ('94022' '94028' '94044' '94011') then pacte_dom ="Plaine_centrale_du_VDM";


/* Département d'entreprise */
if dep_ent not in ('75' '77' '78' '91' '92' '93' '94' '95') then dep_ent_b='99';

else dep_ent_b=dep_ent ;

/* Suppressions blancs à la fin de la variable "pacte" pour enregistrement + tard */
pacte_sitefor1=catt(pacte_sitefor);

pacte_dom1=catt(pacte_dom);

pacte_entp1=catt(pacte_entp);

run;

/* II. Cle@ */
data pacte; set &base;length pacte $100.;

/* Suppression lignes d'avenants */
where TYPE_CONTRAT in ("1" , "2");cont=1;

/* Création modalités de réponses pour variables pacte */
if INSEE_EMP in ('78208' '78423' '78688' '78621' '78356' '78644' '78297') then pacte="SQY";

if INSEE_EMP in ('94022' '94028' '94044' '94011') then pacte="Plaine_centrale_du_VDM";

pacte1=catt(pacte);

if substr(siret_corr,1,1) in (1 , 2) then organisme="public"; else organisme="privé";

/* Création de la variable rupture */
if DATE_RUPTURE ne '' then rupture="oui"; else rupture="non";

run;

/* III. Identification des différents pactes ET garde en mémoire les valeur des pactes*/
proc freq data=pacte ;table pacte1 / out=ttpac;run;

data ttpac; set ttpac; if pacte1 ="" then delete;run;

DATA _null_ ;

     SET ttpac ;

     call symput ("Rgpt"||compress(_N_),pacte1) ;

RUN ;


/*************************************************************************************************/ 


/**

PARTIE 2  Production des différentes sorties statistiques via la macro "sortie_pacte"
**/

/*************************************************************************************************/ 

%macro sortie_pacte_clea ;

%do i=1 %to 2; /* Penser à changer dès que nouveau pacte arrive */
/* Création Variable pour traitements ==> partie 2. */
proc freq data=pacte; table SIRET_CORR*campagne /out=nbre_ent;where pacte="&&Rgpt&i";run;

proc sort data=nbre_ent; by campagne;run;

ods tagsets.excelxp  file="O:\...\&Rgpt&i..xls"  ;

ods listing close;

/* 1. Nbre de contrats signés sur le pacte et évolution 

sur les 4 dernières années + comparaison dpt et IDF */
ods tagsets.excelxp options (sheet_name = 'clea_nbr_cont_dep');

proc tabulate data=pacte;class campagne depent organisme;var cont;

where depent in ('75' '77' '78' '91' '92' '93' '94' '95');

table depent="" all="Région",campagne=""*organisme=""*cont=""*sum="" ;run;

ods tagsets.excelxp options (sheet_name = 'clea_nbr_cont_pacte') ;

proc tabulate data=pacte;class campagne INSEE_EMP organisme;var cont;

where pacte="&&Rgpt&i";

table INSEE_EMP="" all="Pacte",campagne=""*organisme=""*cont=""*sum="" ;run;

/*2. Taux de pénétration de l’apprentissage*/

/* Nombre d'entreprise ayant signé au moins un contrat par campagne sur territoire du pacte */


/* A voir si on récupère les "0 salarié" */
ods tagsets.excelxp options (sheet_name = 'clea_nbr_ent_reg') ;

proc means data=nbre_ent N ; var count ; class campagne ; run;

/*importer table insee avec nbre d entreprise sur le territoire */
/*fair ele rapport entre nb ent insee et apprentissage*/
/*Commune / nb entreprise apprentis / nbre entreprise insee / taux de penetration / (par code insee et total)*/
/*3. Ventilation des établissements accueillant au moins 

apprenti par taille et secteur d’activité*/


/* A. Secteur*/
ods tagsets.excelxp options (sheet_name = 'clea_ent_secteur_pacte') ;

proc tabulate data=pacte;where pacte="&&Rgpt&i"; class campagne a_17;table a_17="" all="total",campagne=""*n="";format A_17 $naf17a. ;run;

ods tagsets.excelxp options (sheet_name = 'clea_ent_secteur_dep');

proc tabulate data=pacte; where depent in ('75' '77' '78' '91' '92' '93' '94' '95'); class campagne a_17 depent;table depent=""*a_17="" all="total",campagne=""*n="";format A_17 $naf17a. ;run;

ods listing;



/* B. Taille*/
ods tagsets.excelxp options (sheet_name = 'clea_ent_taille_pacte');

proc tabulate data=pacte;where pacte="&&Rgpt&i";class campagne eff_corr;table eff_corr="" all="total",campagne=""*n="";format eff_corr ftaille. ;run;

ods tagsets.excelxp options (sheet_name = 'clea_ent_taille_dep');

proc tabulate data=pacte; where depent in ('75' '77' '78' '91' '92' '93' '94' '95'); class campagne eff_corr depent;table eff_corr="" all="total",campagne=""*n="";format eff_corr ftaille. ;run;

/*4. Nbre de rupture sur le pacte et comparaison par rapport 

au département et la région Ile-de-France */


/* A. Nbre de rupture TOTAL */
ods tagsets.excelxp options (sheet_name = 'clea_rupture_pacte');

proc tabulate data=pacte;where pacte="&&Rgpt&i" and campagne="2010/2011" ; 

class rupture INSEE_EMP ;table INSEE_EMP="" all="total",rupture*n="" all="total"; run;

ods tagsets.excelxp options (sheet_name = 'clea_rupture_dep');

proc tabulate data=pacte;where depent in ('75' '77' '78' '91' '92' '93' '94' '95')and campagne="2010/2011" ; 

class rupture depent ;table depent="" all="total",rupture*n="" all="total"; run;

ods listing;



/* B. Nbre de rupture par taille */
ods tagsets.excelxp options (sheet_name = 'clea_rupture_pacte_dep');

proc tabulate data=pacte;where depent in ('75' '77' '78' '91' '92' '93' '94' '95')and campagne="2010/2011" ; 

class rupture depent eff_corr ;table depent="" all="total",eff_corr=""*rupture*n="" all="total"; format eff_corr ftaille. ;run;

ods tagsets.excelxp options (sheet_name = '_clea_rupture_dep_taille');

proc tabulate data=pacte;where pacte="&&Rgpt&i" and campagne="2010/2011" ; 

class rupture INSEE_EMP eff_corr ;table INSEE_EMP="" all="total",eff_corr=""*rupture*n="" all="total"; format eff_corr ftaille. ;run;



/*C. Nbre de rupture par secteur*/
ods tagsets.excelxp options (sheet_name = 'clea_rupture_pacte_secteur');

proc tabulate data=pacte;where pacte="&&Rgpt&i" and campagne="2010/2011" ; 

class rupture INSEE_EMP a_17;table INSEE_EMP="" all="total",a_17=""*rupture*n="" all="total"; format A_17 $naf17a. ;run;

ods tagsets.excelxp options (sheet_name = 'clea_rupture_dep_secteur');

proc tabulate data=pacte;where depent in ('75' '77' '78' '91' '92' '93' '94' '95') and campagne="2010/2011" ; 

class rupture depent a_17 ;table depent="" all="total",a_17=""*rupture*n="" all="total"; format A_17 $naf17a. ;run;

/*4. Nbre d’apprentis résidents sur le territoire Pacte

+ département pour rapport et comparatif */

/* A. Pacte */
ods tagsets.excelxp options (sheet_name = 'gloria_app_résidant_pacte');

proc tabulate data=pacte_gloria;where pacte_dom1="&&Rgpt&i" ;

class CODEPOSTAL;table CODEPOSTAL="" all="Total Pacte",N="Nombre d'apprentis"; run;


/* B. Département */
ods tagsets.excelxp options (sheet_name = 'gloria_app_résidant_dpt');

proc tabulate data=pacte_gloria;class dep_jeune_b;table dep_jeune_b="" all="total",N="Nombre d'apprentis"; format dep_jeune_b $fdep.; run;

/*5. Nbre d’apprentis d’apprentis ayant signé un contrat dans une 

entreprise située sur le territoire du Pacte */

/* A. Pacte */
ods tagsets.excelxp options (sheet_name = 'gloria_entreprise_pacte');

proc tabulate data=pacte_gloria;where pacte_entp1="&&Rgpt&i" ;class CODEPOSTALENT;table CODEPOSTALENT="" all="Total Pacte",N="Nombre d'apprentis"; run;


/* B. Département */
ods tagsets.excelxp options (sheet_name = 'gloria_entreprise_dpt');

proc tabulate data=pacte_gloria;class dep_ent_b;table dep_ent_b="" all="total",N="Nombre d'apprentis"; format dep_ent_b $fdep.; run;

ods listing;

ods tagsets.excelxp close;

%end;

%mend sortie_pacte_clea;

%sortie_pacte_clea

